
Abstract

Mission-time Linear Temporal Logic (MLTL) is rapidly increasing in popularity as a specification
logic, e.g., for runtime verification, model checking, and other formal methods, driving a need for a
larger tool base for analysis of this logic. To that end, we formalize formula progression for MLTL in the
theorem prover Isabelle/HOL. As groundwork, we first formalize the syntax and semantics for MLTL
as well as a verified library of key properties, including useful custom induction rules. We envision
this library as being useful for future formalizations involving MLTL and as serving as a reference
point for theoretical work using or developing MLTL. We then formalize the algorithm and correctness
theorems for formula progression, following the literature. Along the way, we identify and fix several
errors and gaps in the source material. A main motivation for our work is tool validation; we ensure the
executability of our algorithms by using Isabelle’s built-in functionality to generate a code export. This
enables both a formal basis for correctly evaluating MLTL formulas and for automatically generating
provably correct benchmarks for evaluating tools that reason about MLTL.

1 Introduction

Mission-time Linear Temporal Logic (MLTL) [52, 32] adds discrete, closed-interval, integer bounds to
the temporal operators of LTL, providing a finite-trace specification logic that captures many common
requirements of, e.g., embedded systems. MLTL arguably represents the most-used subset of STL [37] and
MTL [2] variants (see [43] for a survey). Accordingly, formalizations of MLTL also serve as foundations
for formalizing those logics.

Labeled timelines are common representations of requirements for operational concepts of aerospace
systems. Due to their comparative ease of use and creation, MLTL has seen wide adoption as a specification
logic in this context. After an extensive survey of verification tools and their associated specification
languages, NASA’s Lunar Gateway Vehicle System Manager (VSM) team selected MLTL as the specification
logic for formalizing their English Assume-Guarantee Contract requirements [14, 15, 16]. The VSM team is
additionally running R2U2 [26], a runtime verification engine that natively reasons over MLTL for on-board
operational verification and on-ground timeline verification. NASA previously used MLTL for specification
of fault disambiguation protocols embedded in the knee of Robonaut2 on the International Space Station
[29], for runtime requirements of NASA’s S1000 octocopter [33] in NASA’s Autonomy Operating System
(AOS) for UAS [34], and for specifying system health management properties of the NASA Swift UAS
[62, 52, 20, 56] and the NASA DragonEye UAS [62, 61]. MLTL was the specification logic of choice for a
UAS Traffic Management (UTM) system [10], an open-source UAS [27], and a high-altitude balloon [35].
JAXA used MLTL for specification of a resource-limited autonomous satellite mission [42]. Other space
systems with MLTL specifications include a CubeSat communications system [36], a sounding rocket [25],

the CySat-I satellite’s autonomous fault recovery system [6], and a case study on small satellites and landers
[55].

Accordingly, many formal methods tools natively analyze MLTL specifications. MLTL was first named
as the input logic to the Realizable Responsive Unobtrusive Unit (R2U2) [52, 58, 26]. The Formal Requirements
Elicitation Tool (FRET) provides a GUI with color-coded segments of structured natural language to elicit
more accurate MLTL specifications from system designers [22, 39, 5]. WEST [18] provides a GUI for
interactive validation of MLTL specifications via regular expressions and sets of satisfying and falsifying
traces. The model checker NuXmv accepts a subset of MLTL for use in symbolic model checking, where
the G and F operators of an LTLSPEC can have integer bounds [30], though bounds cannot be placed
on the U or V (the Release operator of NUXMV). Ogma is a command-line tool to produce monitoring
applications from MLTL formulas [48, 46, 47]. Many more tools accept MLTL specifications without
using that name for their input logics. One example is the Hydra monitoring tool, which advertises that it
reasons about a variation of MTL, though Hydra natively analyzes MLTL specifications as in the case study
[51]. Because MLTL represents a common core of logics used for runtime monitoring, it was selected as
the official logic of the 2018 Runtime Verification Benchmark Competition [57, 1]. We closely examine
one algorithm submitted to generate MLTL benchmarks in that competition via formula progression [31];
later, the Formula PROGression Generator (FPROGG) expanded upon that algorithm to create an automated
MLTL benchmark generation tool [54].

Despite the rapid emergence of tools that analyze MLTL specifications, there is still a shortage of
provably correct formal foundations from which we can validate and verify those tools and their algorithms.
We have SAT solvers for MLTL that work through translations to other logics, proved correct via pencil-
and-paper proofs and experimental validation [32]. There is a native MLTL MAX-SAT algorithm and
implementation that also relies on hand-proofs and experimental demonstrations of consistency [24]. In
contrast, LTL has benefitted from multiple mechanized formalizations, including an Isabelle/HOL library
[65], and a sizeable body of work [63, 66, 64, 60, 19] that has built on this entry to establish a formalized
collection of results for LTL in Isabelle/HOL. There is even a formalizaiton of formula progression for the 3-
valued variant LTL3 [4]. LTL is also formalized in Coq [13] and in PVS, where a library has been developed
to facilitate modeling and verification for systems using LTL [50]. Similarly, MTL is formalized in PVS; this
library was used to verify a translation from a structured natural language to MTL [12, 67].1 A formalization
of MTL in Coq was used to generate OCaml code implementing a past-time MTL monitoring engine [11].
VeriMon [9], a monitoring tool for metric first-order temporal logic (MFOTL) was also developed and
formally verified in Isabelle/HOL using Isabelle’s code export to generate OCaml code for the tool. Notably,
these logics and their associated algorithms can be quite intricate; for example, in the case of Metric Interval
Temporal Logic (MITL), recent work [53] found that the original semantics was incorrectly specified; the
authors develop corrections to some of the original algorithms and avail themselves of formalization in PVS
for an extra layer of trustworthiness.

The time has come for a formalization of MLTL. The remainder of this paper lays out our contributions
as follows. We take a significant step toward the goal of a formal foundation for MLTL by formalizing its
syntax and semantics in the theorem prover Isabelle/HOL [40, 44], as well as an accompanying library of key
properties; see Sect. 2. We develop a formal library establishing key properties and useful lemmas for this
encoding, including duality properties, an NNF transformation, and custom induction rules. We then verify
the formula progression algorithm for MLTL [31] in Sect. 3. Our verified algorithm for formula progression
is executable and can be exported to code in SML, OCaml, Haskell, or Scala, using Isabelle/HOL’s built-in
code exporter. Sect. 4 details the challenges and insights that emerged from our formalization. Our code

1While this library contains some notions relevant for MLTL, it is not specialized for MLTL.

is approximately 3250 lines in Isabelle/HOL and is available on Google drive at: https://drive.
google.com/file/d/1a5YA9wjLOReFZDAEQFXpy1yGHM1dTbpW/view?usp=sharing. We
conclude with a higher-level discussion of the impact and future directions opened up by this work in Sect. 5.

2 Encoding MLTL in Isabelle/HOL

We first overview the syntax and semantics of MLTL, and then discuss our formalization thereof in Isabelle/HOL.
Let AP be a finite set of atomic propositions. The grammar for MLTL formulas is as follows [32, 52]:

ϕ, ψ ::= True | False | p | ¬ϕ | ϕ ∧ ψ | ϕ ∨ ψ | F[a,b] ϕ | G[a,b] ϕ | ϕ U[a,b] ψ | ϕ R[a,b] ψ,

where p ∈ AP, and a, b ∈ Z such that 0 ≤ a ≤ b.
A trace π is a finite sequence of sets of atomic propositions, i.e., π = π[0], π[1], . . . , π[n] where π[i] ⊆

AP for all i ∈ {0, 1, . . . , n}. We refer to each π[i] as a state of the trace π. Throughout the remainder
of the paper, we adopt the following notation from the literature [32, 31]: πk denotes the prefix of trace
π from 0 to k − 1, i.e., πk = π[0], π[1], . . . , π[k − 1], and πk is the suffix of π from k onwards, i.e.,
πk = π[k], π[i+ 1], . . . , π[n]. Each state πi represents the set of atomic propositions that are true at time i.
For easy reference, we visually summarize this notation in Fig. 1.

Figure 1: Prefix and suffix notation for traces.

Trace π satisfies MLTL formula ϕ iff π |= ϕ, where the satisfaction relation |= is defined inductively as
follows [18]:

π |= p iff p ∈ π[0] π |= ¬ϕ iff π ̸|= ϕ

π |= ϕ ∧ ψ iff π |= ϕ and π |= ψ

π |= ϕ ∨ ψ iff π |= ϕ or π |= ψ

π |= F[a,b] ϕ iff |π| > a and ∃i ∈ [a, b] such that πi |= ϕ

π |= G[a,b] ϕ iff |π| ≤ a or ∀i ∈ [a, b] πi |= ϕ

π |= ϕ U[a,b] ψ iff |π| > a and ∃i ∈ [a, b] such that πi |= ψ

and ∀j ∈ [a, i− 1] πj |= ϕ

π |= ϕ R[a,b] ψ iff |π| ≤ a or (∀i ∈ [a, b] πi |= ψ) or
(∃j ∈ [a, b] such that πj |= ϕ and ∀k ∈ [a, j] πk |= ψ)

The temporal operators F,G,U,R are commonly referred to as “Future,” (see, e.g., [49, 68, 59]) “Globally,”
“Until,” and “Release,” respectively.2 MLTL is modeled off of Linear Temporal Logic (LTL) [52]; like

2The literature sometimes uses ♢ and □ to denote the Future and Globally operators respectively, but we use F and G as a
stylistic choice to match the other temporal operators. The Future operator is also sometimes called “Finally” in the literature
[32, 18].

LTL, there is also the option of including more temporal operators in the syntax of MLTL, such as a Next
operator and variations on Until and Release, which are easily defined in terms of the above grammar.
We design our syntax with a view towards formalizing some current MLTL algorithms, particularly the
formula progression algorithm [32] and the MLTL-to-regular expressions algorithm [18], which are defined
on the above grammar. However, in any development that necessitates frequent use of additional temporal
operators, it would be straightforward to expand our core syntax either by directly creating an alternative
expanded version of the syntax or by defining abbreviations for these additional temporal operators. We
present our syntactical encoding of MLTL in Sect. 2.1.

We follow the mathematical presentation of MLTL in our formalization; in particular, we define each
of the temporal operators F,G,U,R and then formalize important properties of these operators. For this,
we mostly follow existing pencil-and-paper proofs [18], deviating as necessary to make the proofs work in
Isabelle/HOL. Sect. 2.2 overviews our formalization of MLTL semantics, and Sect. 2.3, Sect. 2.4 present our
formalization of basic properties and useful functions for MLTL, respectively.

Although our development is standalone in that it does not import anything other than the standard HOL
library, we drew some inspiration from the formalization of a related logic, Linear Temporal Logic (LTL),
in Isabelle/HOL [65].

2.1 Syntax

Our syntactic encoding captures the standard formula shapes. For atomic formulas, we allow True (in
True mltl), False (False mltl), and atomic propositions (Prop mltl) that take a variable with arbitrary
type as an argument (’a is Isabelle’s arbitrary type); we use the arbitrary type to provide maximum flexibility
to users of our framework. We allow the usual logical connectives: ¬ (Not mltl), ∧ (And mltl), and
∨ (Or mltl). Lastly, we allow a standard set of temporal operators: future (Future mltl), globally
(Global mltl), until (Until mltl), and release (Release mltl); each temporal operator takes two
natural numbers to specify their associated time bounds. Putting it all together, we have the following
datatype in Isabelle/HOL to represent MLTL formulas:

datatype (atoms_mltl: ’a) mltl =
True_mltl | False_mltl | Prop_mltl ’a

| Not_mltl "’a mltl" | And_mltl "’a mltl" "’a mltl"
| Or_mltl "’a mltl" "’a mltl"
| Future_mltl "’a mltl" "nat" "nat"
| Global_mltl "’a mltl" "nat" "nat"
| Until_mltl "’a mltl" "’a mltl" "nat" "nat"
| Release_mltl "’a mltl" "’a mltl" "nat" "nat"

Note that our syntax does not include the standard well-definedness assumption on intervals for temporal
operators. For instance, our syntax allows the formula “Future mltl True mltl 5 3”, even though this
is not a well-defined formula in MLTL. We get around this by explicitly requiring that these intervals have
well-defined bounds in our semantics and also by adding this as an assumption in our correctness theorems
when necessary.

2.2 Semantics

Because traces are always finite in MLTL, we encode traces as lists of sets of variables of arbitrary type (in
Isabelle/HOL, this is type ’a set list). Fig. 2 illustrates our representation of traces in Isabelle/HOL
with an example.

Figure 2: Here, we have two Boolean variables, a and b, and four timesteps. At time 0, variable a is true
(and b is false), at time 1, both a and b are true, and so on. In Isabelle, we would represent this trace as the
list [{a}, {a, b}, {b}, {a}].

We define the semantics of MLTL formulas in Isabelle/HOL in the function semantics mltl, which
takes a trace π and an MLTL formula ϕ and returns true if π |= ϕ and false otherwise. Our formal definition
mirrors the mathematical definition, but occasionally makes explicit assumptions that are typically implicit.
For example, in the semantics of atomic propositions, we explicitly ensure that if π |= Prop mltl p

holds, then π is nonempty (for well-definedness). Additionally, in the semantics of temporal operators, we
explicitly ensure that the associated time bounds are well-defined (i.e, a ≤ b).

We now present our formal semantics for a representative subset of operators (Prop mltl, And mltl,
Global mltl, and Until mltl); the full semantics is in Appendix A.1.

fun semantics_mltl::"[’a set list, ’a mltl] ⇒ bool"
where

| "semantics_mltl π (Prop_mltl q) =
(π ̸= [] ∧ q ∈ (π ! 0))"

| "semantics_mltl π (And_mltl φ ψ) =
(semantics_mltl π φ ∧ semantics_mltl π ψ)"

| "semantics_mltl π (Global_mltl φ a b) =
(a ≤ b ∧ (length π≤a ∨ (∀i::nat. (i≥a ∧ i≤b)
−→ semantics_mltl (drop i π) φ)))"

| "semantics_mltl π (Until_mltl φ ψ a b) =
(a≤b ∧ length π>a ∧ (∃i::nat. (i≥a ∧ i≤b) ∧
(semantics_mltl (drop i π) ψ ∧ (∀j. j≥a ∧ j<i
−→ semantics_mltl (drop j π) φ))))"

The semantics for Prop mltl checks that the atomic proposition q is in the set of atomic propositions at
the initial timestep (in Isabelle/HOL, this is π ! 0, as ! is the operator to access an element from a list), while
the semantics for And mltl checks that trace π satisfies both subformulas φ, ψ.

The semantics for Global mltl checks that the subformula φ is satisfied at every step in the trace.
More precisely, (Global mltl φ a b) holds automatically if the length of φ is less than a; otherwise,
for all i between a and b, we must have (drop i π) |= φ. Here, (drop i π) encodes πi, the suffix of
the trace starting at timestep i, as drop is Isabelle/HOL’s operator to drop the first i elements from a (zero-
indexed) list. Note that our definition for the semantics of Global mltl includes a well-definedness check
to ensure that a≤b.

In other words, the semantics for Until mltl checks that the subformula ψ is satisfied at some point
in the future, and that the subformula ϕ is satisfied at every timestep up to that point. We again check a≤b
for well-definedness, and then we check that there exists an i between a and b such that (drop i π) |=
ψ, and that φ is satisfied at all timesteps up to i, i.e., (drop j π) |= φ for all j between a and i-1.

5



We highlight that the expressiveness of Isabelle/HOL allows us to state the semantics of MLTL formulas
essentially verbatim from the mathematical definition, adding in implicit assumptions (like a≤b in the
semantics for the temporal operators) when necessary.

Formalizing the semantics in this way further clarifies the differences between MLTL and closely-
related logics. Perhaps the most closely-related is MTL-over-naturals [43], with which MLTL differs in five
important ways: the traces are finite, the intervals are finite, the internals are closed and unit-less (generic),
signal processing is compartmentalized, and the U -semantics obeys the interval. Note that the U -semantics
of MTL-over-naturals is: π |= ϕ U[a,b]ψ iff |π| > a and, ∃i ∈ [a, b], i < |π| such that π, i |= ψ and
∀j < i, it holds that π, j |= ϕ. Bounded LTL and its variants (e.g., [38]) share this divergent U -semantics.
Note that other variants of MTL, such as Metric Interval Temporal Logic (MITL) [3] treat the temporal
operator intervals as continuous (rather than discrete) intervals that may be open-ended over real numbers
(not integers), and even prohibit singular time intervals, of the form [a, a]. Bounded LTL (BLTL) [28] and
LTLf [21] both reason over finite traces like MLTL, but share syntax with LTL; they have (unspecified)
finite bounds on the time domain rather than intervals on each temporal operator and keep the X operator.
(LTLf adds a weak next operator as well.)

Throughout our development, we found it useful to have a definition of semantic equivalence for MLTL
formulas, where two formulas are semantically equivalent if and only if they are satisfied by the same set of
traces. In Isabelle/HOL, we define this as:

definition semantic_equiv:: "’a mltl ⇒ ’a mltl ⇒ bool"
where "semantic_equiv φ ψ = (
if (∀π. (semantics_mltl π φ)=(semantics_mltl π ψ))
then True else False )"

This function, semantic equiv, tests if φ and ψ are equivalent by checking if their semantics match on all
traces π.

2.3 Properties

To further establish the correctness of our definitions, and to establish their future usability, we formally
prove various essential properties of our encoding of MLTL. Our aim is to establish a reusable formalized
library of MLTL properties; this library is useful for both developing future verified tools and algorithms
based on MLTL, and as a reference point for theoretical developments involving MLTL.

We first consider properties involving relationships and equivalences between operators. Whereas we
explicitly define each core temporal operator (Future, Globally, Until, Release) in terms of its mathematical
definition, an alternative approach is to formally define all operators in terms of a functionally complete set
of operators for MLTL, like {¬,∧,U} [32], and then derive the typical mathematical definitions for the other
operators. This approach is common in the literature [32, 59, 24], and is a primarily stylistic difference. For
organizational purposes, we prefer to have all of the mathematical definitions in one place and all of the
properties in a separate file. To ensure that our encoding matches the alternative approach, we state and
prove key equivalences between operators.

Many of the following equivalences prove easily with Isabelle/HOL’s built-in automation tactics. For
example, we establish that Future can be rewritten with Until in the lemma future as until by showing
that F[a,b]ϕ is equivalent to True U[a,b]ϕ, and we establish the duality of Future and Globally (i.e., that G[a,b]ϕ
is equivalent to ¬F[a,b]¬ϕ) in globally future dual.3 In Isabelle/HOL, we have:

3Both of these properties are commonly used in the literature to define Future and Globally in terms of Until.

lemma future_as_until:
fixes a b::"nat"
assumes "a ≤ b"
shows "semantic_equiv (Future_mltl φ a b)

(Until_mltl (True_mltl) φ a b)"

lemma globally_future_dual:
fixes a b::"nat"
assumes "a ≤ b"
shows "semantic_equiv (Global_mltl φ a b)

(Not_mltl (Future_mltl (Not_mltl φ) a b))"

These two lemmas, alongside DeMorgan’s Laws for ∧ and ∨, are almost enough to establish that
{¬,∧, U} is functionally complete for MLTL. We also need to prove the duality of U and R, i.e., that
ϕ R[a,b]ψ is equivalent to ¬(¬ϕ U[a,b]¬ψ). In Isabelle, we state this in the release until dual lemma:

lemma release_until_dual:
fixes a b::"nat"
assumes a_leq_b: "a ≤ b"
shows "semantic_equiv (Release_mltl φ ψ a b)
(Not_mltl (Until_mltl (Not_mltl φ) (Not_mltl ψ) a b))"

This proof does not follow immediately with Isabelle’s automation and was more involved. We follow a
proof sketch from the literature [18]. The biconditional statement is split into two implications, and in
each direction, the proof considers three cases: 1⃝ when the length of the trace is less than a, 2⃝ when
ψ is satisfied for all timesteps in the interval [a, b], and 3⃝ when ψ is not satisfied for some timestep in the
interval [a, b]. We follow the source material [18] in all cases except for the forward direction of 3⃝; here, we
find an error in the source material, which incorrectly reduces a subgoal of showing that ∀s ∈ [a, b].(πs |=
ψ ∨ (∃t ∈ [a, s − 1]. πt |= ϕ)) to showing that ∀s ∈ [a, b].(∃t ∈ [a, s − 1]. πt |= ϕ). Instead, we directly
prove the subgoal with Sledgehammer [45], and then the rest of the proof closed with definition unrolling.

These lemmas are sufficient to establish that any MLTL formula can be rewritten using only the operators
{¬,∧,U}, which is defined in the literature [32] as the Backus Naur Form (BNF) for MLTL. We verify a
custom induction rule, bnf induct, which allows us to perform structural induction on MLTL formulas in
BNF; in Isabelle/HOL, this is:

lemma bnf_induct[case_names IntervalsWellDef PProp
True False Prop Not And Until]:

assumes IntervalsWellDef: "intervals_welldef η"
and PProp: "(

∧
η φ. (semantic_equiv η φ −→ P η = P φ))"

and True: "P True_mltl"
and False: "P False_mltl"
and Prop: "

∧
p. P (Prop_mltl p)"

and Not: "
∧

η φ. [[η = Not_mltl φ; P φ]] =⇒ P η"
and And: "

∧
η φ ψ. [[η = And_mltl φ ψ;

P φ; P ψ]] =⇒ P η"
and Until:"

∧
η φ ψ a b.

[[η = Until_mltl φ ψ a b; P φ; P ψ]] =⇒ P η"
shows "P η"

In order to apply this induction rule, we have to prove the IntervalsWellDef and PProp properties.
IntervalsWellDef makes explicit an assumption that is typically left implicit: If the MLTL formula η that

7



we are trying to perform structural induction on contains any temporal operators, then their corresponding
time intervals should be well-defined (for example, if η is “Until mltl a b True mltl False mltl”,
then a must be less than or equal to b); we formalize this in our intervals welldef function.4 PProp

stipulates that the property P that we are trying to prove about MLTL formulas is invariant on semantically
equivalent formulas: i.e., for any semantically equivalent formulas η and φ, P holds on η (denoted P η) if
and only if P holds on φ.

Given both IntervalsWellDef and PProp, the induction rule allows us to conclude P η if we
can prove property P on MLTL formulas of the shapes η = True mltl, η = False mltl, and η =

Prop mlt p (these are the base cases of the induction rule), and prove the property P holds on formulas
of the shapes η = Not mltl φ, η = And mltl φ ψ, and η = Until mltl φ ψ, given the appropriate
inductive hypotheses for each case. When it applies, then, this induction rule is impactful as it considerably
reduces the number of cases when compared to the default structural induction proof rule for MLTL formulas
(which splits into cases on each operator in our syntax).

We do not limit ourselves to establishing the standard functional completeness for MLTL, but prove a
number of additional useful basic properties, like distributivity properties for the Until and Release operators
over the logical connectives ∧ and ∨. Most of these prove easily with Isabelle’s built-in automation tactics.

2.4 Useful Functions and Custom Induction Rules

We further flesh out our encodings of the standard MLTL definitions and equivalences to create a reusable
library of useful functions on MLTL formulas. Our first useful function is called convert nnf; it converts
an MLTL formula to its negation normal form (NNF), i.e., a logically equivalent formula where negations
are only in front of atomic propositions. This function successively applies the operator duality properties
to push negations down to the level of atomic propositions. We prove that converting a formula to NNF
preserves the semantics of the formula, and also that convert nnf is idempotent in the following Isabelle/HOL
lemmas:

lemma convert_nnf_preserves_semantics:
assumes "intervals_welldef φ"
shows "semantic_equiv φ (convert_nnf φ)"

lemma convert_nnf_convert_nnf:
shows "convert_nnf (convert_nnf φ) = convert_nnf φ"

The proofs of these two lemmas work by induction on the depth of the formula; accordingly, we
formalize a depth function for MLTL formulas called depth mltl. We also define a function called
nnf subformulas to compute the set of subformulas of an MLTL formula, and prove (again by induction
on the depth of the formula) that any subformula of a formula in NNF is also in NNF.

lemma nnf_subformulas:
assumes "φ = convert_nnf init_φ"
assumes "ψ ∈ subformulas φ"
shows "∃ init_ψ. ψ = convert_nnf init_ψ"

Using these functions and properties, we establish the following custom induction rule, called nnf induct,
as another useful tool for proving properties about MLTL formulas. While the bnf induct rule reduces the
number of cases to consider when proving properties about MLTL formulas, the nnf induct rule allows

4For the curious reader, we present this function in Appendix A.2.

us to consider simpler cases when proving properties about MLTL formulas in NNF. In particular, this
induction rule requires the property P to hold on formulas of the each of the standard shapes (True, False,
Prop, And, Or, Future, Globally, Until, Release), but limits the Not case to only consider negations of atomic
propositions (in case Not Prop). This induction rule imposes no additional requirements on the property P
(as opposed to the bnf induct rule, which requires P to be invariant on semantically equivalent formulas),
but does impose the restriction that our input formula η is in NNF form (in assumption nnf). Fortunately, all
MLTL formulas can be efficiently converted to NNF [18], so this assumption can often be easily satisfied.

lemma nnf_induct[case_names nnf
True False Prop And Or Final
Global Until Release NotProp]:

assumes nnf: "∃φ. η = convert_nnf φ"
and True: "P True_mltl"
and False: "P False_mltl"
and Prop: "

∧
p. P (Prop_mltl p)"

and And: "
∧
η φ ψ.

[[η = And_mltl φ ψ; P φ; P ψ]] =⇒ P η"
and Or: "

∧
η φ ψ.

[[η = Or_mltl φ ψ;P φ; P ψ]] =⇒ P η"
and Final: "

∧
η φ a b.

[[η = Future_mltl φ a b; P φ]] =⇒ P η"
and Global: "

∧
η φ a b.

[[η = Global_mltl φ a b; P φ]] =⇒ P η"
and Until: "

∧
η φ ψ a b.

[[η = Until_mltl φ ψ a b; P φ; P ψ]] =⇒ P η"
and Release: "

∧
η φ ψ a b.

[[η = Release_mltl φ ψ a b; P φ; P ψ]] =⇒ P η"
and Not_Prop: "

∧
η p. η =

Not_mltl (Prop_mltl p) =⇒ P η"
shows "P η"

Finally, we formalize the computation length of an MLTL formula. The computation length, complen,
of an MLTL formula is recursively defined as follows, where p ∈ AP is a atomic proposition, and ϕ and ψ
are MLTL formulas [18]:

complen(p) = complen(¬p) = 1,

complen(ϕ ∧ ψ) = complen(ϕ ∨ ψ) =
max(complen(ϕ),complen(ψ)),

complen(G[a,b]ϕ) = complen(F[a,b]ϕ) = b+ complen(ϕ),

complen(ϕ U[a,b] ψ) = complen(ϕ R[a,b] ψ) =

b+max(complen(ϕ)− 1,complen(ψ))

In Isabelle/HOL, we define the complen mltl function, which maps an MLTL formula of arbitrary
type to a natural number, to capture this notion. The definition is essentially verbatim, modulo the requisite
syntax changes, except that we add the definition for computation length of True mltl and False mltl

(both are equal to 1), which were missing from the literature [18]. We present a few representative cases of
the formal definition here; the full formal definition is in Appendix A.

fun complen_mltl:: "’a mltl ⇒ nat"
where "complen_mltl (Prop_mltl p) = 1"
| "complen_mltl (And_mltl φ ψ) =

max (complen_mltl φ) (complen_mltl ψ)"
| "complen_mltl (Global_mltl φ a b) =

b + (complen_mltl φ)"
| "complen_mltl (Until_mltl φ ψ a b) =

b + (max ((complen_mltl φ)-1) (complen_mltl ψ))"

Computation length is a slightly optimized version of an earlier and closely-related notion of the worst-
case propagation delay (wpd) of a formula. Worst-case propagation delay was introduced in the context
of runtime verification [29, Definition 5] and was designed to give the maximum number of timesteps an
observer needs to wait from the current timestep to decide satisfaction of an MLTL formula so that the
decision will not change with further information.5 For example, if we try to evaluate the satisfaction
of F[0,2]p on a trace of length 2, then it is possible that further information could change the verdict.
Specifically, F[0,2]p is false on the trace [{}, {}] but true on the trace [{}, {}, {p}]. However, after complen
(F[0,2]p) = 3 timesteps, the value of p in subsequent timesteps will not change whether the trace satisfies
F[0,2]p. In particular, the R2U2 tool [29], which checks satisfaction of MLTL formulas on traces, does not
return an answer until it is certain that the answer will not change with further information. In the worst
case, the duration of this delay from the current time is captured by the wpd of the formula.

Interestingly, we found that the computation length also plays a key role in the formal proofs of the
formula progression algorithm. Although the source material which we were following [31] does not use
or mention computation length, we found that this notion directly fixes some issues with the proofs. We
now turn to our formalization of formula progression; we will further discuss computation length and some
useful properties we formalize thereof in Sect. 3.4.

3 Formula Progression

Formula progression is a common technique that steps through time to evaluate formulas; it dates back to
the 1990’s [7] and has been developed for various temporal logics including MTL [17], LTLf [41], and
MLTL [31]. Formula progression represents a straightforward and generally easy-to-implement algorithm
for evaluating whether a particular temporal trace satisfies the given formula, and is therefore utilized in
many contexts, including satisfiability solving, validation, runtime verification, and benchmark generation.

Armed with our encoding of MLTL, we discuss our formalization of the MLTL formula progression
algorithm and the associated correctness theorems [31]. While we mostly follow the original paper [31]
that developed formula progression for MLTL, we found and fixed several errors in the top-level correctness
results. In particular, the source material states three key theorems (and a corollary); Fig. 3 overviews these
top-level results and our changes to them. We now discuss the formula progression algorithms and each
top-level result in turn, with an emphasis on our modifications.

5The definitions of computation length and wpd are mostly the same. The main difference is that computation length includes
a -1 term in the Until and Release operators; this is not present in wpd and it represents a small optimization. Another difference
is that the computation length of atomic propositions is defined to be 1, while the wpd of atomic propositions is defined to be 0,
but this difference is purely cosmetic, as the wpd captures the delay in the number of timesteps from the current timestep, as is the
standard in runtime verification, while computation length captures the length of the total.

Statement Additions / Modifications Lines 
(Source)

LOC 
(Isabelle)

Theorem 1: decomposition
 prog(φ, 𝜋) = prog( prog(φ, 𝜋k), 𝜋k)

Filled in previously elided proof of the following:
   prog(prog(φ, 𝜋k ), 𝜋 [𝑘]) = prog(𝜙, 𝜋k+1) 

6 ~135

Theorem 2: satisfiability preserving
  𝜋 ⊨ φ ⟷ 𝜋k ⊨ prog(φ, 𝜋k)

Modified assumption on k from:
   1 ≤ k ≤ |𝜋|     to     1 ≤ k < |𝜋| 5 ~750

Theorem 3: correctness
  𝜋 ⊨ φ ⟷ (prog(φ, 𝜋) ≡ True)

Added key assumption on computation length:
   complen(φ) ≤ |𝜋| 6 ~1000

Corollary 1: trace extension
  𝜋 ⊨ φ → 𝜋 @ ζ ⊨ φ

Added key assumption on computation length:
   complen(φ) ≤ |𝜋| 3 ~15

Figure 3: We summarize the top-level results for formula progression and our modifications thereof and
compare the number of lines in the proof sketches in the source material [31] with the corresponding LOC
in Isabelle/HOL. As our formal proof of Theorem 3 required identifying and formalizing many properties
of the computation length, we count these in the LOC.

3.1 Encoding the Formula Progression Algorithm

Formula progression takes an MLTL formula and steps through a trace, partially evaluating the formula
at each state in the trace. At each timestep, formula progression partially evaluates the input formula and
transforms it into a simplified formula which, under some generally unrestrictive conditions, is logically
equivalent. Because MLTL is an inherently bounded logic, it is well-suited for formula progression; one can
easily predetermine the number of timesteps needed to transform the input formula into True or False.

The definitions for MLTL formula progression [31, Definition 1] reflect this compatibility, and they are
straightforward to formalize. For a trace π of length greater than 1, prog(φ, π) is defined as prog(prog(φ, π[0]),
φ1), which we formalize as the function formula progression:

fun formula_progression::
"’a mltl ⇒ ’a set list ⇒ ’a mltl"
where "formula_progression φ π =

(if length π = 0 then φ
else (if length π = 1 then

(formula_progression_len1 φ (π!0))
else (formula_progression

(formula_progression_len1 φ (π!0)) (drop 1 π))))"

This function takes an MLTL formula with atoms of arbitrary type and a trace (a list of sets containing
elements of the same arbitrary type). To evaluate prog(F, π[0]), it cases on the length of π. If π is empty,
it returns F; if π has length 1, it calls the helper function formula progression len1, which handles
formula progression on traces of length 1. When π has length longer than 1, it evaluates prog(F, π[0]) as
formula progression len1 F (π!0)—here, π!0 accesses the element at index 0 from π—and then
passes the result to formula progression on drop 1 π, where drop n L is Isabelle’s function to drop
the first n elements from list L.

The helper function formula progression len1 cases on the structure of the input formula, closely
following the associated mathematical definitions [31, Definition 1]. As a representative example, we show
one of the most complicated cases, for formulas of the shape ϕ U[a,b] ψ. Mathematically,

prog(ϕ U[a,b] ψ, π) =


ϕ U[a−1,b−1] ψ if 0 < a ≤ b
prog(ψ, π) ∨
(prog(ϕ, π) ∧
ϕ U[0,b−1] ψ) if 0 = a < b

prog(ψ, π) if 0 = a = b.

This is encoded in Isabelle/HOL as follows:

formula_progression_len1 (Until_mltl φ ψ a b)
k = (if (0<a∧a≤b) then (Until_mltl φ ψ (a-1) (b-1))

else (if (0=a∧a<b) then
(Or_mltl (formula_progression_len1 ψ k)

(And_mltl (formula_progression_len1 φ k)
(Until_mltl φ ψ 0 (b-1))))
else (formula_progression_len1 ψ k)))

Aside from notational differences, the only difference between this formal function and the mathematical
definition is a minor technical one: the type of formula progression len1 is ’a mltl ⇒’a set

⇒’a mltl, where we use ’a set for the second argument instead of ’a set list because we are (for
convenience) directly passing in the singleton element in our trace π of length one, which we call k = π!0.
We provide the full formalized function formula progression len1 in Appendix B.

We demonstrate the executability of our formalized functions by utilizing Isabelle’s code generation
feature to evaluate the formula progression of the formula G[0,2]p on the trace [{p}, {p}, {}]. We visualize the
computations involved in this example in Figure 4; here, we are using the value command in Isabelle/HOL
(which invokes the code generator [23]) to evaluate intermediate steps of the formula progression algorithm.
It is important to note that the formula progression algorithm is not guaranteed to produce an output in the
simplest form; our verified implementation produces a final output of ¬((¬True) ∨ ((¬True) ∨ (¬False))),
which is logically equivalent to False.

3.2 Verifying Theorem 1

The first key theorem for formula progression informally states that performing formula progression along
a trace can be split into first performing formula progression on the prefix, and then on the suffix, of that
trace. Mathematically, it states that prog(ϕ, π) = prog(prog(ϕ, πk), πk) for 1 ≤ k ≤ length π. In
Isabelle, we prove the following theorem, formula progression decomposition, where “take k π”
is the function to take the first k elements of the list π (lists are zero-indexed, so this is the prefix of π from
0 to k− 1, which is πk) and “drop k π” is the function to drop the first k elements of the list π (which is
πk; cross-reference Fig. 1).

theorem formula_progression_decomposition:
fixes φ::"’a mltl"
assumes "k ≥ 1"
assumes "k ≤ length π"
shows "formula_progression

(formula_progression φ (take k π)) (drop k π)
= formula_progression φ π"

For Section 4.1
prog  (G[0,2] p0)
  [{p0}, {p0}, { }]

prog (prog_len1 (G[0,1] p0) {p0}) 
                [{ }]

prog (G[0,1] p0)   
          [{p0}, { }]

prog (prog_len1 (G[0,2] p0) {p0}) 
       [{p0}, { }]

prog (G[0,0] p0)   
                 [{ }]

prog_len1 (G[0,0] p0) { } False

Figure 4: We illustrate the recursive calls that the formula progression algorithm makes to transform the
formula G[0,2]p on the trace [{p}, {p}, {}] to the formula False. The formula requires that p is true at every
timestep from 0 to 2; in timesteps 0 and 1, p is true, and thus the time bound is reduced from [0, 2] to [0, 1] and
then to [0, 0]. In the final timestep, p is false (since the state is empty and does not contain p), and so formula
progression ultimately returns False. Note that the output of formula progression is not guaranteed to be in
the simplest form; in this example, our verified implementation produces¬((¬True)∨((¬True)∨(¬False))),
which is logically equivalent to False.

Our formal proof of this theorem follows the proof sketch in the original source paper [31] relatively
closely.

We found it effective to split out the following identity used in the proof into a separate helper lemma:
prog(prog(ϕ, πk), π[k]) = prog(ϕ, πk+1).

Although the source paper stated that this property holds by definition, we proceeded to prove this by
a more involved induction on k. This highlights the role of formalization, which necessitates making all
handwaving in source proofs rigorous. Even so, Theorem 1 was the easiest of the three top-level formula
progression theorems to prove, and its proof only required about 135 LOC (compared to 6 lines of proof in
the source material).

3.3 Verifying Theorem 2

The second key theorem for formula progression states that (in most cases) the formula progression of a
formula on a trace is logically equivalent to the formula progression of the prefix of the trace on its suffix.
Mathematically, this is π |= ϕ iff πk |= prog(ϕ, πk); the source material [31] stated this theorem for
1 ≤ k ≤ length π, but our formal statement is for 1 ≤ k < length π. In Isabelle/HOL, we have:

theorem satisfiability_preservation:
fixes φ::"’a mltl"
assumes "k ≥ 1"
assumes "k < length π"
assumes "intervals_welldef φ"
shows "semantics_mltl (drop k π)

(formula_progression φ (take k π))
←→ semantics_mltl π φ"

In addition to altering the bound on k, we also add a well-definedness assumption (intervals welldef

φ) which asserts that all temporal operators have well-defined timebounds (e.g., ifφ is Release mltlψ η a b,

then a ≤ b). This well-definedness assumption is typically included as part of the MLTL syntax; it is an
artifact of formalization that our syntax does not enforce this well-definedness and it must be made explicit
here (cross-reference Sect. 2.1).

The altered bound, however, is mathematically necessary for the correctness of the result and was
overlooked in the source material. We uncovered this while formalizing the base case of this theorem.
The source material inducts on k and states that the base case (where k = 1) is by induction without
providing details; which caused issues when working in Isabelle/HOL. The issue we run into here is related
to end-of-trace behavior. When both k = 1 and the length of π is 1, the statement of Theorem 2 becomes
π |= ϕ←→ [] |= prog(ϕ, π), and this does not hold in general.

Specifically, consider a formula of the shape G[0,b]p for b > 0 with trace π = [{p}] of length 1. If we
allow k = 1, then we would want to establish π |= G[0,b]p ←→ π1 |= prog(G[0,b]ϕ, π

1). We have that
π1 = [] and π1 = π, so this becomes π |= G[0,b]p ←→ [] |= prog(G[0,b]p, π). Then using the semantics
of MLTL on the left-hand side, π |= G[0,b]p reduces to ∀i ∈ [0, b]. πi |= p, which further reduces to
∀i ∈ [0, b].(πi ̸= [] ∧ p ∈ πi[0]). However, this is false because π1 = [].

Applying the definition of formula progression to the right-hand side, we have that:

prog(G[0,b]p, π) = ¬prog(F[0,b]¬p, π)
= ¬(¬prog(p, π) ∨ F[0,b−1]¬p)
= ¬(¬true ∨ F[0,b−1]¬p)
= ¬F[0,b−1]¬p
= G[0,b−1]p , Future and Globally are duals

Thus, the right-hand side becomes [] |= G[0,b−1]p; using the semantics of MLTL this is |[]| ≤ 0 ∨ (∀i ∈
[0, b− 1]. []i |= p), which is true because the left disjunct is true.

Fortunately, modifying the statement of Theorem 2 to insist k < length π instead of k ≤ length π
removes this issue (because, for k = 1, our traces π must have length at least 2, so we do not encounter the
above issues with the empty trace). Further, and crucially, we are still able to use our modified version of
Theorem 2 in the proofs of Theorem 3 and the top-level corollary; that is, our modification fixes Theorem 2
without overly weakening the result.

3.4 Verifying Theorem 3 and Corollary 1

The third key theorem for formula progression informally states that, for a sufficiently long input trace π,
and for a well-defined input formula φ, π models formula φ if and only if the formula progression of φ over
π is logically equivalent to True. More precisely, we require (in the first assumption) that all intervals in φ
are well-defined, and we require (in the second assumption) that the length of π is at least the computation
length of the formula (cross-reference Sect. 2.4, [18]). In Isabelle, we have the following theorem, where
semantic equiv encodes logical equivalence of two MLTL formulas (cross-reference Sect. 2.4):

theorem formula_progression_correctness:
fixes φ::"’a mltl"
assumes "intervals_welldef φ"
assumes "length π ≥ complen_mltl φ"
shows "(semantic_equiv (formula_progression φ π)

True_mltl) ←→ semantics_mltl π φ"

While the assumption that all intervals in φ are well-defined can again be viewed as something of a
formalization curiosity as it is typically mathematically implicit, the assumption that the length of π is
greater than or equal to the computation length of the formula φ is mathematically crucial for the correctness
of the theorem. Notably, this assumption was missing in the corresponding theorem in the original source
material [31, Theorem 3]; identifying it is a contribution of our formalization. Fortunately, adding in this
assumption does not affect the practicality of formula progression; in practice, traces are typically long
streams of data, so it is not overly restrictive to assume that a trace is sufficiently long.

We identified this assumption when attempting to prove the base case of Theorem 3. The source material
states that this is by structural induction and omitted all mathematical details. When initially attempting to
formalize the base case without this critical assumption, we quickly ran into issues in the Or mltl case. To
see why, let us consider the proof on a mathematical level. We are trying to prove that (π |= ϕ ∨ ψ) ←→
(prog(ϕ ∨ ψ, π) = True) for a trace π of length 1, given that (π |= ϕ) ←→ (prog(ϕ, π) = True) and
(π |= ψ) ←→ (prog(ψ, π) = True). Because prog(ϕ ∨ ψ, π) is defined as prog(ϕ, π) ∨ prog(ψ, π) when
π has length 1, and using that π |= ϕ ∨ ψ ←→ (π |= ϕ ∨ π |= ψ), our goal becomes (π |= ϕ ∨ π |=
ψ) ←→ ((prog(ϕ, π) ∨ prog(ϕ, π)) = True). Then, using our inductive hypotheses, we incur the goal:
(prog(ψ, π) = True ∨ prog(ϕ, π) = True) ←→ ((prog(ϕ, π) ∨ prog(ψ, π)) = True). This does not hold;
prog(ϕ, π) ∨ prog(ψ, π) can be logically equivalent to True without either prog(ϕ, π) or prog(ψ, π) being
logically equivalent to True; all that is required is that prog(ϕ, π) must hold in any states where prog(ψ, π)
is false and vice-versa.

By adding the assumption on the computation length, the base case of the formula progression theorem
inherits the (structurally strong) assumption that 1 is greater than or equal to the computation length of
the input formula ϕ. This assumption is strong enough to establish that the formula progression of ϕ on
any trace of length 1 is either globally true or globally false. This solves the above issue in the Or mltl

case; since (prog(ϕ, π) and (prog(ψ, π) are now globally either True or False, we can indeed conclude
(prog(ψ, π) = True ∨ prog(ϕ, π) = True) ←→ ((prog(ϕ, π) ∨ prog(ψ, π)) = True), as desired. We
establish this useful property of the computation length in the following key lemma:

lemma complen_bounded_by_1:
fixes π::"’a set list"
assumes "intervals_welldef φ"
assumes "1 ≥ complen_mltl φ"
shows "(∀ η. semantics_mltl η

(formula_progression_len1 φ (π!0)) = True)
∨ (∀ η. semantics_mltl η
(formula_progression_len1 φ (π!0)) = False)"

This lemma is sufficient to prove our strengthened base case; however, we also need to ensure that
the strengthened base case is usable in the proof of Theorem 3. This requires adding several additional
properties of the computation length. In particular, we establish the following two crucial properties:

lemma complen_one_implies_one:
assumes "intervals_welldef φ"
assumes "complen_mltl φ = 1"
shows "complen_mltl (formula_progression φ π) = 1"

lemma formula_progression_decreases_complen:
assumes "intervals_welldef φ"
shows "complen_mltl φ = 1 ∨

complen_mltl (formula_progression φ π) = 1 ∨

complen_mltl (formula_progression φ π)
≤ complen_mltl φ - (length π)"

This first lemma, complen one implies one, proves that (for formulas φwith well-defined intervals),
if the computation length of φ is 1, then the computation length of the formula progression of φ is also 1; i.e.,
formula progression does not increase the computation length for formulas that already have the minimum
computation length. The second key lemma, formula progression decreases complen, establishes
that performing formula progression on a formula φ over a trace π usually decreases the computation length
of φ proportionally to the length of π. More precisely, either the computation length of φ equals 1 (i.e.,
is already minimal), or the computation length of the formula progression of φ on trace π equals 1 (i.e.,
is minimal), or the computation length of the formula progression of φ on π is less than or equal to the
computation length of φ minus the length of the trace π (i.e., the computation length of the progression is
decreased by the length of π).

We use complen one implies one in the proof of formula progression decreases complen,
and we use both of these lemmas in the proof of Theorem 3 to establish that complen prog(ϕ, (πlength π−1)) ≤
length(π length π−1), which is equivalent to complen prog(ϕ, (πlength π−1)) ≤ 1. This then allows us to use
the base case of Theorem 3 with the formula prog(ϕ, (πlength π−1)) and the trace π length π−1 to prove that
prog(prog(ϕ, (πlength π−1), π length π−1)) = True is equal to π length π−1 |= prog(ϕ, (πlength π−1)). Then,
following the source material [31], we can conclude that 1⃝ prog(prog (ϕ, (πlength π−1), π length π−1)) =
prog(ϕ, π) using Theorem 1 and 2⃝ π length π−1 |= prog(ϕ, (πlength π−1)) iff π |= ϕ using Theorem 2. Putting
these pieces together closes the proof.

In total, formula progression correctness took approximately 1000 lines of code to formalize
(this includes about 800 lines for establishing the necessary properties of the computation length), as
compared to 6 lines in the original source material.

Putting the three key theorems together, we achieve the following top-level corollary, which informally
states that, for a well-defined MLTL formula φ and a sufficiently long trace π, if π |= φ, then any trace
whose prefix is π also models φ. Formally, we prove the following result in Isabelle/HOL, where @ is
Isabelle/HOL’s syntax for appending lists.

corollary trace_append:
fixes φ::"’a mltl"
assumes "intervals_welldef φ"
assumes "length π ≥ complen_mltl φ"
assumes "semantics_mltl π φ"
shows "semantics_mltl (π @ ζ) φ"

The source material [31, Corollary 1] had stated this corollary without the key assumption on the length
of the trace, but that is incorrect. To see why, consider the following example. We know that [{2}] |=
G[1,3]False, as the length of the trace [{2}] is ≤ 1 (cross-reference the mathematical semantics of MLTL
presented in Sect. 2). From this, without the assumption that [{2}] must be greater than or equal to the
computation length of G[1,3]False, which is 4, we could obtain, for example, [{2}, {3}, {4}, {2, 3}] |=
G[1,3]False, which contradicts the semantics of MLTL. Fortunately, after adding in this assumption, the
(formal) proof of this corollary is quite straightforward and follows the proof presented in the source material
[31] without issue.

After formalizing this corollary, we realized that we could go one step further to also formalize an
alternate version of Theorem 3: If a sufficiently long trace π does not satisfy a formula ϕ, then the formula
progression of ϕ on π is logically equivalent to False. Although the proof of this exactly mirrors the proof of

Theorem 3, it is not immediately implied by Theorem 3; assuming that π ̸|= ϕ, Theorem 3 only yields that
the prog(ϕ, π) is not semantically equivalent to True. This does not necessitate that the resulting formula is
semantically equivalent to False.

Combining the original and alternate versions of Theorem 3, we obtain the following important property
of formula progression:

lemma formula_progression_true_or_false:
fixes φ::"’a mltl"
assumes "intervals_welldef φ"
assumes "length π ≥ complen_mltl φ"
assumes "ψ = (formula_progression φ π)"
shows "(semantic_equiv ψ False_mltl)

∨ (semantic_equiv ψ True_mltl)"

This allows us to conclude that for a sufficiently long trace π, the formula progression of ϕ on π is either
logically equivalent to True or False. In fact, our earlier result
complen bounded by 1 is a special case of this more general property.

We also use our alternate version of Theorem 3 to prove the inverse of the top-level corollary, strengthening
it to an if-and-only-if statement:

corollary trace_append_iff:
fixes φ::"’a mltl"
assumes "intervals_welldef φ"
assumes "length π ≥ complen_mltl φ"
shows "semantics_mltl π φ ←→

(∀ ζ. semantics_mltl (π @ ζ) φ)"

This says that for a sufficiently long trace π, extending π with any additional states (in ζ) does not affect
the satisfaction of formula ϕ. This formalizes the intuition for computation length, that timesteps beyond
the computation length of a formula do not affect the satisfaction of that formula. Retrospectively, one
bonus of our formalization is that it nicely integrates various related concepts (like formula progression and
computation length) for MLTL in a single centralized and easily extensible library.

4 Challenges and Insights

We view our formalization as emblematic of the potential benefits of formalizing source material that
heavily relies on intuition. The mathematical intuition for formula progression for MLTL is clear because
of its boundedness. Accordingly, the original source material heavily leaned into this intuition, and many
important details were elided (possibly in light of space constraints), which introduced bugs. In particular,
the base cases of Theorem 2 and Theorem 3 claimed to hold by structural induction; mathematically, this
seems quite intuitive. However, formalization serves as a useful forcing function, where all intuition must
be made precise, and in the course we found that the base cases were incorrect as stated and needed
modifications; these then propagated into the top-level theorems and corollary.

One of the main creative challenges that we faced was identifying the necessary assumptions to add to
these two base cases. These added assumptions needed to walk a fine line. On the one hand, they must be
strong enough to prove the base cases. On the other hand, they must be weak enough that they can still be
used to prove the desired top-level result. Additionally, we wanted our additional assumptions to maintain
the practicality of formula progression. We found that it was useful to keep the high-level context in mind

throughout the course of our formalization; we frequently switched back and forth between working on a
base case and using that base case to prove the overall desired result.

Similarly, it was only by analyzing the top-level corollary that we identified the main necessary assumption
for Theorem 3 (that length π ≥ complen mltl φ). Mathematically, we found it intuitively clear that this
assumption would be sufficient to establish the corollary and that it would not affect the practicality of
formula progression (because traces in real-world systems tend to be long). Interestingly, though, we were
initially unsure whether it was strong enough to fix the formalization of the proof of Theorem 3. Establishing
this required identifying and proving various key properties of the computation length, particularly those
discussed in Sect. 3, which establish that formula progression tends to decrease computation length. Initially,
we stated these in a series of (unproven) properties of the computation length that we wanted to be able to
use in our proofs (as a sort of wishlist); we were pleasantly gratified when we later found that all of these
necessary properties held.

Another key challenge that we encountered was related to end-of-trace behavior. We noticed this
especially in the proof of Theorem 2, where the behavior of the empty trace caused issues in the theorem as
it was originally stated. In practice, MLTL traces are typically long, and we suspect this is why it is easy
to overlook the edge cases where we are reasoning about the empty trace. Additionally, MLTL end-of-trace
behavior is somewhat unintuitive; for example, [] |= G[0,1]False according to MLTL semantics, because
Globally is satisfied if the length of the trace is less than or equal to the lower bound on its associated
interval. Fortunately, formalization is excellent at making strange edge cases precise and ensuring that they
are correctly reasoned about.

Finally, we wish to comment on our choice of the theorem prover Isabelle/HOL, which we benefited
from throughout the formalization. We particularly appreciate the automation afforded by Sledgehammer
[45] and also the built-in support for unrolling definitions and functions; the latter was extremely helpful
in a number of our proofs that rely on structural induction and reduce into detailed casework.6 We also
appreciate Isabelle’s well-developed libraries; though our work does not depend on any entries from the
Archive of Formal Proofs, we loosely modeled our formalization of MLTL syntax and semantics on an
existing formalization of LTL [65].

5 Discussion and Future Work

Our work serves to refine and unify the previously-published semantics, algorithms, and definitions for
MLTL. In particular, as the crown jewel of our formalization, we verify the formula progression algorithm
[31]; along the way, we fix some errors in the correctness proofs. We also develop the foundational notion
of computation length [18]. In general, we found that the source material was somewhat piecemeal in its
development of this notion of computation length and its counterpart, worst-case propagation delay [29].
One contribution of our formalization is establishing a centralized body of lemmas about the computation
length, and it is an exciting area of future work to continue to develop this corpus. We can now work
toward establishing that the computation length is minimal, and proving related results from the literature,
particularly ones involving worst-case propagation delay, e.g. [69, Lemma 4].

Our formalization of the syntax, semantics, duality properties, an NNF transformation, custom induction
rules, and a formula progression algorithm constitutes a useful library that facilitates future formal developments
for MLTL and serves as a useful reference point for theoretical developments. In particular, we envision a

6Most notably, in the base cases of Theorems 2 and 3, the subcases for the temporal operators Until, Future, Release, and
Globally are very involved, as the formula progression function for these operators on a trace of length one splits into different
cases.

suite of formally verified algorithms for MLTL that can be used to validate the existing (unverified) tools
for analyzing MLTL formulas, or which could be developed into tools in their own right. We ensure the
executability of our algorithms and generate a code export so that they can be used to validate existing
(unverified) tool support, i.e., the FPROGG tool, which implements formula progression in C [54]. We
envision creating a conformance tester for MLTL, similar to the success of VeriMon for related modal
logics [8]. In this way, our library can also serve to create provably-correct benchmarks for evalation of
MLTL satisfiability and runtime verification. A common reason why competitions like the 2018 attempt
to start a Runtime Verification Competition [1] fail is due to a lack of a large, challenging, and correct set
of benchmarks. Such formalizations also serve to validate existing tool implementations. For example, the
PVS formalization of MTL was created to verify a translation from a structured natural language called
FRETish to MTL in order to validate the tool FRET [12]. Now the Isabelle/HOL formalization of MLTL
can serve a similar role in validating all of the tools that natively analize MLTL and other logics for which
it is a subset.

Since MLTL is a commonly-used subset of other popular logics like STL and MTL, our library also
lays a foundation for their future formalization in Isabelle/HOL. In particular, a natural next step would be
utilizing our library in a formalization of the syntax, semantics, and related lemmas for MTL-over-naturals.
From there, it will be possible to keep building formalizations of other MTL variants, such as the list in [43].
The library we have created here will serve as a necessary component of any future efforts to formalize more
complex semantics that extend MLTL, such as STL.

A Appendix A

In this appendix, we present some of the code snippets for functions that we did not include (or only partially
included) in the main body of the paper. These may also be found in our code, but we include them here for
easy reference.

A.1 MLTL Semantics

We present our full formal definition of the semantics of MLTL (cross-reference Sect. 2).

fun semantics_mltl::"[’a set list, ’a mltl] ⇒ bool"
where
"semantics_mltl π True_mltl = True"

| "semantics_mltl π False_mltl = False"
| "semantics_mltl π (Prop_mltl q) =

(π ̸= [] ∧ q ∈ (π ! 0))"
| "semantics_mltl π (Not_mltl φ) =

(¬ (semantics_mltl π φ))"
| "semantics_mltl π (And_mltl φ ψ) =

(semantics_mltl π φ ∧ semantics_mltl π ψ)"
| "semantics_mltl π (Or_mltl φ ψ) =

(semantics_mltl π φ ∨ semantics_mltl π ψ)"
| "semantics_mltl π (Future_mltl φ a b) =

(a≤b ∧ length π>a ∧ (∃i::nat.
(i≥a ∧ i≤b) ∧ semantics_mltl (drop i π) φ))"

| "semantics_mltl π (Global_mltl φ a b) =
(a ≤ b ∧ (length π≤a ∨ (∀i::nat. (i≥a ∧ i≤b)
−→ semantics_mltl (drop i π) φ)))"

| "semantics_mltl π (Until_mltl φ ψ a b) =
(a≤b ∧ length π>a ∧ (∃i::nat. (i≥a ∧ i≤b) ∧
(semantics_mltl (drop i π) ψ ∧ (∀j. j≥a ∧ j<i

−→ semantics_mltl (drop j π) φ))))"
| "semantics_mltl π (Release_mltl φ ψ a b) =

(a≤b ∧ (length π≤a ∨ (∀i::nat. (i≥a ∧ i≤b) −→
(semantics_mltl (drop i π) ψ)) ∨
(∃j. j≥a ∧ j≤b-1 ∧ semantics_mltl (drop j π) φ
∧ (∀k. a≤k ∧ k≤j −→
semantics_mltl (drop k π) ψ))))"

This semantics is designed to closely match the mathematical semantics of MLTL while also including
the well-definedness assumption for temporal operators (that a ≤b).

A.2 Well-Defined Intervals

We present our Isabelle/HOL function intervals welldef, which checks if all intervals associated to
temporal operators in an input formula are well-defined.

fun intervals_welldef:: "’a mltl ⇒ bool"
where "intervals_welldef True_mltl = True"
| "intervals_welldef False_mltl = True"
| "intervals_welldef (Prop_mltl p) = True"
| "intervals_welldef (Not_mltl φ) = intervals_welldef φ"
| "intervals_welldef (And_mltl φ ψ) =
(intervals_welldef φ ∧ intervals_welldef ψ)"
| "intervals_welldef (Or_mltl φ ψ) =
(intervals_welldef φ ∧ intervals_welldef ψ)"
| "intervals_welldef (Future_mltl φ a b) =
(a ≤ b ∧ intervals_welldef φ)"
| "intervals_welldef (Global_mltl φ a b) =
(a ≤ b ∧ intervals_welldef φ)"
| "intervals_welldef (Until_mltl φ ψ a b) =
(a ≤ b ∧ intervals_welldef φ ∧ intervals_welldef ψ)"
| "intervals_welldef (Release_mltl φ ψ a b) =
(a ≤ b ∧ intervals_welldef φ ∧ intervals_welldef ψ)"

This function is recursively defined. It returns True on formulas that clearly have no temporal operators
(True mltl, False mltl, and Prop mltl). For formulas of shapes And mltl, Not mltl, and Or mltl,
intervals welldef checks whether the relevant subformulas contain well-defined intervals. For example,
for And mltl φ ψ, this function checks whether the intervals in φ and ψ are well-defined. For the
temporal operators, intervals welldef checks whether the top-level interval is well-defined and also
whether the intervals of any relevant subformulas are well-defined. For example, for intervals welldef

Future mltl φ a b to hold, we must have both a ≤ b and also intervals welldef φ.

A.3 Computation Length

We present our full encoding of the comptuation length of an MLTL formula complen mltl (cross-
reference Sect. 2.4).

fun complen_mltl:: "’a mltl ⇒ nat"
where "complen_mltl False_mltl = 1"
| "complen_mltl True_mltl = 1"
| "complen_mltl (Prop_mltl p) = 1"

| "complen_mltl (Not_mltl φ) = complen_mltl φ"
| "complen_mltl (And_mltl φ ψ) =

max (complen_mltl φ) (complen_mltl ψ)"
| "complen_mltl (Or_mltl φ ψ) =

max (complen_mltl φ) (complen_mltl ψ)"
| "complen_mltl (Global_mltl φ a b) =

b + (complen_mltl φ)"
| "complen_mltl (Future_mltl φ a b) =

b + (complen_mltl φ)"
| "complen_mltl (Release_mltl φ ψ a b) =

b + (max ((complen_mltl φ)-1) (complen_mltl ψ))"
| "complen_mltl (Until_mltl φ ψ a b) =

b + (max ((complen_mltl φ)-1) (complen_mltl ψ))"

B Appendix B

We detail the full formula progression function, and we present the correspoding encoding of the corresponding
function in Isabelle, formula progression len1 (cross-reference Sect. 3.1).

Let π be a trace and ϕ and ψ be MLTL formulas. The formula progression of ϕ on π, denoted
prog(ϕ, π), is defined recursively as follows [31, Definition 1]:

• If |π| = 1, then

– prog(True, π) = True and prog(False, π) = False;

– if ϕ = p is an atomic proposition,
prog(p, π) = True if and only if p ∈ π[0];

– prog(¬ϕ, π) = ¬prog(ϕ, π);
– prog(ϕ ∨ ψ, π) = prog(ϕ, π) ∨ prog(ψ, π);
– prog(ϕ ∧ ψ, π) = prog(ϕ, π) ∧ prog(ψ, π);

– prog(ϕU[a,b]ψ, π) =


ϕ U[a−1,b−1] ψ if 0 < a ≤ b;
prog(ψ, π) ∨
(prog(ϕ, π) ∧
ϕ U[0,b−1] ψ) if 0 = a < b;

prog(ψ, π) if 0 = a = b;

– prog(F[a,b]ϕ, π) =


F[a−1,b−1]ϕ if 0 < a ≤ b;
prog(ϕ, π)∨
F[0,b−1]ϕ if 0 = a < b;

prog(ϕ, π) if 0 = a = b;

– prog(ϕR[a,b]ψ, π) = ¬prog((¬ϕ)U[a,b](¬ψ), π);
– prog(G[a,b]ϕ, π) = ¬prog(F[a,b](¬ϕ), π).

• Else, prog(ϕ, π) = prog(prog(ϕ, π[0]), π1).

The |π| > 1 case is described in detail in the Isabelle function formula progression in Sect. 3.1. We
present the full |π| = 1 case as the function formula progression len1 in Isabelle as follows:

function formula_progression_len1::
"’a mltl ⇒ ’a set ⇒ ’a mltl" where
"formula_progression_len1 True_mltl k =

True_mltl"
| "formula_progression_len1 False_mltl k =

False_mltl"
| "formula_progression_len1 (Prop_mltl p) k =

(if p∈k then True_mltl else False_mltl)"
| "formula_progression_len1 (Not_mltl φ) k =

Not_mltl(formula_progression_len1 φ k)"
| "formula_progression_len1 (And_mltl φ ψ) k =

And_mltl (formula_progression_len1 φ k)
(formula_progression_len1 ψ k)"

| "formula_progression_len1 (Or_mltl φ ψ) k =
Or_mltl (formula_progression_len1 φ k)

(formula_progression_len1 ψ k)"
| "formula_progression_len1 (Until_mltl φ ψ a b)
k = (if (0<a∧a≤b) then (Until_mltl φ ψ (a-1) (b-1))

else (if (0=a∧a<b) then
(Or_mltl (formula_progression_len1 ψ k)

(And_mltl (formula_progression_len1 φ k)
(Until_mltl φ ψ 0 (b-1))))
else (formula_progression_len1 ψ k)))"

| "formula_progression_len1 (Release_mltl φ ψ a b)
k = Not_mltl (formula_progression_len1
(Until_mltl (Not_mltl φ) (Not_mltl ψ) a b) k)"

| "formula_progression_len1 (Global_mltl φ a b) k =
Not_mltl (formula_progression_len1
(Future_mltl (Not_mltl φ) a b) k)"

| "formula_progression_len1 (Future_mltl φ a b) k =
(if 0<a∧a≤b then (Future_mltl φ (a-1) (b-1))
else if (0=a∧a<b) then
(Or_mltl (formula_progression_len1 φ k)
(Future_mltl φ 0 (b-1)))
else (formula_progression_len1 φ k))"

This function exactly captures the structural cases of the formula progression function on traces of
length 1. The Future case splits into cases very similar to the cases that Until splits into, and the Release
and Globally cases utilize the dual properties to utilize the Until and Future cases, respectively. The
clear correspondence between the mathematical definition and the Isabelle encoding is a testament to the
expressiveness of Isabelle/HOL and its ease of use for formalization.

